

Yankee Shipping has received a request from VDog company to ship dog food to their customers in Boston, Newark and Toronto. VDog has warehouses in Chicago and Detroit. VDog requests Yankee Shipping to arrange for transportation at the lowest possible cost. The shipping cost between origins and destinations is provided as following:

Shipping Cost	Boston	Newark	Toronto
Chicago	\$4,000	\$2,500	\$1,500
Detroit	\$3,000	\$1,500	\$1,000

The dog food demand for Boston, Newark and Toronto is 500, 1500 and 1000 lbs respectively. The maximum inventory (supply) for Chicago and Detroit is 2000 and 1500 respectively.

Find the amount to be transported between each warehouse and customers at a minimum and at a maximum cost.

Defining Sets:

Let $DC = \{C', D'\}$ be the distribution centers

Let $WH = \{B', N', T'\}$ be the warehouses

Defining Inputs:

Let $C_{DC,WH} = [4000, 2500, 1500; 3000, 1500, 1000]$ be cost of shipping from DC to WH

Let $D_{WH} = [500, 1500, 1000]$ be the demand at WH

Let $S_{DC} = [2000, 1500]$ be the supply at DC

Defining Variables:

Let $X_{DC,WH} = \{1 \text{ if we ship anything from DC to WH}, 0 \text{ otherwise}\} \Rightarrow \text{binary}$

Let $Y_{DC,WH} = \text{AMOUNT shipped from DC to WH} \geq 0$

Defining Objective:

Min Cost = $\sum_{DC} \sum_{WH} C_{DC,WH} \times X_{DC,WH}$

Defining Constraints

$\sum_{WH} Y_{DC,WH} \leq S_{DC} \quad \forall DC$

$\sum_{DC} Y_{DC,WH} \geq D_{WH} \quad \forall WH$

$Y_{DC,WH} \leq S_{DC} \times X_{DC,WH} \quad \forall DC, WH$

```

proc optmodel;
set<str> DC={'C','D'};
set<str> WH={'B','N','T'};

num C{DC,WH} = [4000 2500 1500
                 3000 1500 1000];

num D{WH} = [500 1500 1000];
num S{DC} = [2000 1500];

var X{DC,WH} binary;
var Y{DC,WH}>=0;

min Cost = sum{a in DC, b in WH} (C[a,b]*X[a,b]);
con c1{a in DC}: sum{b in WH} (Y[a,b])<=S[a];
con c2{b in WH}: sum{a in DC} (Y[a,b])>=D[b];
con c3{a in DC, b in WH}: Y[a,b] <= S[a]*X[a,b];

solve;

print cost;
print X;
print y;
quit;

```